

COMP1216. Software Modelling and Design (2021-22)

Group 21: Vaccine Centre Modelling in Event-B

Submission date: 13 March 2022

Contents

1	Introduction	3
2	Task 1. Class diagram	4
3	Task 2. Event-B model 1.1 User Management System	5
4	Task 2. Event-B model 1.2 Vaccination Centres & 1.4 Receiving Vaccines	11
5	Task 2. Event-B model 1.3 Appointment booking and rebooking	18

1 Introduction

Assignment summary

The objective is to create a formal modelling of a COVID Vaccination Tracking System. The system is designed to encompass the entirety of the journey from the creation of the vaccination centre to the administering the vaccination itself and giving the user a certificate as proof. This is broken down into four sections:

- The user management system.
- Vaccination centres.
- Appointment Booking and Re-booking
- Receiving vaccines

Task Distribution

Alin was tasked with creating the class diagram and assisted on sections 2 and 3.

Bagir assisted on sections 1 and 2 and consultant on code commenting.

Daniel was the main architect on sections 1, 2 and 3. Also tasked with code commenting.

Logan was tasked with reformulating the class diagram, and producing the LaTeX document.

2 Task 1. Class diagram




Figure 1: Class Diagram

3 Task 2. Event-B model 1.1 User Management System

The user management system falls under 1.1 of the assignment. The objective was to create a set of registered users that could log in and out of a system using a password. Administrators and staff to be designed as special users where the system can only start with an administrator who is a root user.

REQ 1: *The system manages a set of registered users* - Here users(citizens), admins (administrators) and staff are all subsets of the USER type.

REQ 2: *The system should only allow a user to log in to the system using the correct password* - The system checks if the user belongs to the domain of registered users, checks that the password belongs to the registered user and ensure that the user does not belong to the domain of already logged in users.

REQ 3: *A logged-in user can log out of the system* - The user is checked against the domain of logged in users and that the password belongs to the user. If these conditions are met it logs the user out.

REQ 4: *There are three types of users: administrators, staff, and citizens* - The system has three user types as requested. As referenced in REQ 1, the users (citizens), admins (administrators) and staff are all subsets of the USER type.

REQ 5: *A user can only have one role in the system* - Each user (users, admins, staff) have a clearly defined role with separate capabilities within the system. This is checked by an invariant where the intersection of the users, admins and staff is equal to the empty set and modelled by removing and assigning the users to/from these sets.

REQ 6: *The root user is the initial administrator* - It was decided to use a separate event - InitRootUser to initialize the root user. The event is available only when there are no administrators in the system and initialises one.

REQ 7: *A logged-in administrator can register a new user for the role of administrator or staff*

The event ElevateUserStaff can register upgrade a user to a staff role. It verifies that the user belongs to the domain of administrators to permit the elevation, checks that the user to be elevated belongs to the domain of users and they are logged in. There is a union between the elevated_user and the staff and thereafter the elevated_user is removed from users as the upgrade to staff is complete.

There are two other events, ElevateStaffAdmin which elevates a staff member to an admin role, and ElevateUserAdmin which elevates a user to an admin role. These function similarly to the ElevateUserStaff event.

REQ 8: *A citizen can register an account with the system directly.* - The event RegisterUser permits for the user to register an account directly with the system. There are checks to ensure that the user does not belong to the domain of previously registered users and that the

cardinality of the administrators is greater than zero (making sure the system is initialized first).

REQ 9: *A logged-in user can change their password* - The users, admins and staff have their own events to change their passwords. For the citizens it is `ChangerUserPassword`, for the staff the event is `ChangeStaffPassword` and for the administrators the event is `ChangeAdminPassword`. It was necessary to have different events for each to update the correct set in the event.

```
1 context UserManagementContext
2 sets
3   USER
4   PASSWORD
5 end
```

```
1 machine UserManagement
2 sees UserManagementContext
3
4 /* User Management machine that deals with user registration, login, roles and role management.
5 * Requirements implemented:
6 * REQ 1 The system manages a set of registered users
7 * REQ 2 The system should only allow a user to log in to the system using the correct password
8 * REQ 3 A logged-in user can log out of the system
9 * REQ 4 There are three types of users: administrators, staff, and citizens
10 * REQ 5 A user can only have one role in the system
11 * REQ 6 The root user is the initial administrator
12 * REQ 7 An logged-in administrator can register a new user for the role of administrator or staff
13 * REQ 8 A citizen can register an account with the system directly.
14 * REQ 9 A logged-in user can change their password
15 */
16
17 variables
18   users
19   admins
20   staff
21   login
22   register
23
24 invariants
25 /* Users can have the same passwords. A many to one relationship*/
26 @inv-user-password: users ∈ USER ↔ PASSWORD
27 @inv-admin-password: admins ∈ USER ↔ PASSWORD
28 @inv-staff-password: staff ∈ USER ↔ PASSWORD
29
30 /* Users, admins and staff are all part of the USER type*/
31 @inv-users-user: dom(users) ⊆ USER
32 @inv-admins-user: dom(admins) ⊆ USER
33 @inv-staff-user: dom(staff) ⊆ USER
34
35 /* Users can have only one role */
36 @inv-types-no-intersection: users ∩ admins ∩ staff = ∅
37
38 /* Registered users are a union of all roles */
```

```

39 @inv-login-registered: register = users ∪ admins ∪ staff
40
41 /* Logged in users is a subset of registered users */
42 @inv-login-sub-registered: login ⊆ register
43
44 events
45
46 /* Initialize all sets */
47 event INITIALISATION
48 then
49   @act-init-users: users := ∅
50   @act-init-admins: admins := ∅
51   @act-init-staff: staff := ∅
52   @act-init-login: login := ∅
53   @act-init-register: register := ∅
54 end
55
56 /* Initialize the root user as admin if there are no admins */
57 event InitRootUser
58 any
59   root
60   pass
61 where
62   @grd1: root ∈ USER
63   @grd2: pass ∈ PASSWORD
64   @check-no-admins: card(admins) = 0
65 then
66   @assign-root: admins := {root ↦ pass}
67   @assign-register: register := {root ↦ pass}
68 end
69
70 /* Register a new basic user */
71 event RegisterUser
72 any
73   user
74   pass
75 where
76   @grd1: user ∈ USER
77   @grd2: pass ∈ PASSWORD
78   @check-unregistered: user ∉ dom(register)
79   @check-nonzero-admin: card(admins) > 0
80 then
81   @add-register: register := register ∪ {user ↦ pass}
82   @add-users: users := users ∪ {user ↦ pass}
83 end
84
85 /* Login a user */
86 event LoginUser
87 any
88   user
89   pass
90 where
91   @check-registered: user ∈ dom(register)
92   @check-password: pass = register(user)
93   @check-not-loggedin: user ∉ dom(login)
94 then

```

```

95   @add-login: login := login  $\cup$  {user  $\mapsto$  pass}
96   end
97
98   /* Logout a user */
99   event LogoutUser
100  any
101    user
102    pass
103  where
104    @check-loggedin: user  $\in$  dom(login)
105    @check-password: pass = login(user)
106  then
107    @remove-login: login := login  $\setminus$  {user  $\mapsto$  pass}
108  end
109
110  /* Change a user's password */
111  event ChangeUserPassword
112  any
113    user
114    pass
115    new_pass
116  where
117    @check-user: user  $\in$  dom(users)
118    @check-password: pass = login(user)
119    @new-password-type: new_pass  $\in$  PASSWORD
120    @check-loggedin: user  $\in$  dom(login)
121  then
122    /** Change the password for user everywhere */
123    @act1: login := login  $\triangleleft$  {user  $\mapsto$  new_pass}
124    @act2: register := register  $\triangleleft$  {user  $\mapsto$  new_pass}
125    @act3: users := users  $\triangleleft$  {user  $\mapsto$  new_pass}
126  end
127
128  /* Change a staff's password */
129  event ChangeStaffPassword
130  any
131    user
132    pass
133    new_pass
134  where
135    @check-staff: user  $\in$  dom(staff)
136    @check-password: pass = login(user)
137    @new-password-type: new_pass  $\in$  PASSWORD
138    @check-loggedin: user  $\in$  dom(login)
139  then
140    /** Change the password for user everywhere */
141    @login-change: login := login  $\triangleleft$  {user  $\mapsto$  new_pass}
142    @register-change: register := register  $\triangleleft$  {user  $\mapsto$  new_pass}
143    @staff-change: staff := staff  $\triangleleft$  {user  $\mapsto$  new_pass}
144  end
145
146  /* Change an admin's password */
147  event ChangeAdminPassword
148  any
149    user
150    pass

```



```

151  new_pass
152  where
153    @check-admin: user ∈ dom(admins)
154    @check-password: pass = login(user)
155    @new-password-type: new_pass ∈ PASSWORD
156    @check-loggedin: user ∈ dom(login)
157  then
158    /** Change the password for user everywhere */
159    @login-change: login := login ◁ {user ↦ new_pass}
160    @register-change: register := register ◁ {user ↦ new_pass}
161    @admin-change: admins := admins ◁ {user ↦ new_pass}
162  end
163
164  /* Elevate a user to staff role */
165  event ElevateUserStaff
166  any
167    user
168    pass
169    elevate_user
170  where
171    @check-admin: user ∈ dom(admins)
172    @check-password: pass = login(user)
173    @check-elevate-from-user: elevate_user ∈ dom(users)
174    @check-loggedin: user ∈ dom(login)
175  then
176    @add-to-staff: staff := staff ∪ ({elevate_user} ◁ users)
177    @remove-from-users: users := users \ ({elevate_user} ◁ users)
178  end
179
180  /* Elevate a user to admin role */
181  event ElevateUserAdmin
182  any
183    user
184    pass
185    elevate_user
186  where
187    @check-admin: user ∈ dom(admins)
188    @check-password: pass = login(user)
189    @check-elevate-from-user: elevate_user ∈ dom(users)
190    @check-loggedin: user ∈ dom(login)
191  then
192    @add-to-admins: admins := admins ∪ ({elevate_user} ◁ users)
193    @remove-from-users: users := users \ ({elevate_user} ◁ users)
194  end
195
196  /* Elevate staff member to admin role */
197  event ElevateStaffAdmin
198  any
199    user
200    pass
201    elevate_user
202  where
203    @check-admin: user ∈ dom(admins)
204    @check-password: pass = login(user)
205    @check-elevate-from-staff: elevate_user ∈ dom(staff)
206    @check-loggedin: user ∈ dom(login)

```

```
207 then
208   @add-to-admins: admins := admins  $\cup$  ({elevate_user}  $\triangleleft$  staff)
209   @remove-from-staff: staff := staff  $\setminus$  ({elevate_user}  $\triangleleft$  staff)
210 end
211
212 end
213
214
```

4 Task 2. Event-B model 1.2 Vaccination Centres & 1.4 Receiving Vaccines

Several parameters were given for section 1.2:

- An administrator must be able to create a vaccination centre.
- After the centre's creation any administrator can provide it with updates.
- These updates are vaccine stock and appointments.
- Vaccines must have brands and belong to a unique batch number.
- Citizens must have a name, unique NHS number and receive a certificate after receiving their vaccination.
- A vaccination centre has vaccine stock and a number of available vaccines.

From a standpoint of the system it was easier to fulfill REQ 30 and REQ 31 as part of the vaccination centre, as this is where the citizen both receives the vaccine and the certificate is administered.

For how each requirement was addressed please see below:

REQ 10: *A vaccine has a brand and a batch number* - The set vaccines define a batch number to brand relationship. The event addVaccine is created such that it checks if a brand is a valid brand and that the batch number to be added is unique. The batch must not be a member of the domain of pre-existing vaccines, this is how it remains unique. The event also assigns a number of vaccines for the batch.

REQ 11: *A citizen has a name, a unique NHS number and a list of certificates for received shots* - nhs_numbers is the set defining the user and nhs number relation. Citizen_names links each nhs number to a name. Later, using their NHS number, the certificates of a citizen can be retrieved..

REQ 12: *Each certificate includes the vaccine information and the date of receiving the shot* - The set certificate_dates stores the date of each certificate and certificate_vaccines stores the batch number of the vaccine administered. Using that information it is possible to find all the other information about the vaccine like the centre and brand.

REQ 13: *A citizen can retrieve their certificates from the system* - The event GetCitizenCertificateUser lets a user who is already logged in request their certificates. The result is stored in the citizen_certs set.

REQ 14: *A vaccination centre has a vaccine stock specifying the availability of the vaccines* - center_stock stores the batch number to centre relation. Using the batch number one can then retrieve the vaccine count in that batch from vaccines_count to check the vaccine availability of the center.

REQ 15: *An administrator can update the vaccine stock by adding more vaccine batches* - AssignBatchToCenter event enables the administrator to assign a batch to a centre by adding it to the center_stock. The event guards check that the batch is not assigned to another centre already and that the batch number is valid.

REQ 16: *A vaccination centre has a daily number of appointments* - The set appointments_nr

stores a relation from centres to an integer number. Admins can change that number by updating that relation with a new number.

REQ 17: *A vaccination centre maintains a list of booked appointments* - appointment_center stores appointments of type APPOINTMENT linked to a centre. We can query the appointments for a centre by providing a centre as input.

REQ 18: *An administrator or a staff can view the citizen details using their NHS number* - The event GetCitizenCertificatesNHS enables an administrator who is already logged into the system to retrieve the user's certificates. Other information could be easily queried by inputting the NHS number into other relations of the system.

REQ 30 & 31: *A staff member can update the citizen's certificates after administering the vaccine (REQ 30) & the vaccine stock needs to be updated when the certificate is produced for a citizen (REQ 31)* - The IssueCertificate event is defined in machine 2 and extended with 2 more guards in machine 3. We make sure that a certificate can be issued only by a logged in staff member. The NHS number is associated with a certificate showing that a vaccine was administered and the vaccines_count set is updated by decrementing the vaccine count related to the batch number of the vaccine administered.

```
1 context VaccinationCentersContext extends UserManagementContext
2 sets
3   BRAND
4   BATCH_NR
5   NAME
6   NHS_NR
7   CENTER
8   CERTIFICATE
9 end
```

```
1 machine VaccinationCenters refines UserManagement
2 sees VaccinationCentersContext
3
4 /* Vaccination Centers
5  * Requirements implemented:
6  * REQ 10 A vaccine has a brand and a batch number
7  * REQ 11 A citizen has a name, a unique NHS number and a list of certificates for received shots
8  * REQ 12 Each certificate includes the vaccine information and the date of receiving the shot
9  * REQ 13 A citizen can retrieve their certificates from the system
10 * REQ 14 A vaccination centre has a vaccine stock specifying the availability of the vaccines
11 * REQ 15 An administrator can update the vaccine stock by adding more vaccine batches
12 * REQ 16 A vaccination centre has a daily number of appointments
13 * REQ 17 An administrator or a staff can view the citizen details using their NHS number
14 * REQ 30 A staff member can update the citizens certificates after administering the vaccine.
15 */
16
17 variables
18   users admins staff login register
```

```

19 vaccines
20 centers
21 center_stock
22 center_stock_amount
23 center_vaccines
24 day
25 citizen_names
26 nhs_numbers
27 certificates
28 appointments_nr
29 citizen_name
30 vaccines_count
31 citizen_certs
32 certificate_vaccines
33 certificate_dates
34
35 invariants
36 @typeof-centers: centers  $\subseteq$  CENTER
37 @typeof-vaccines: vaccines  $\in$  BATCH_NR  $\leftrightarrow$  BRAND
38 @typeof-vaccines-count: vaccines_count  $\in$  BATCH_NR  $\leftrightarrow$   $\mathbb{Z}$ 
39 @center_stock-total_func: center_stock  $\in$  BATCH_NR  $\leftrightarrow$  CENTER
40 @center_vaccines: center_vaccines  $\subseteq$  BATCH_NR
41 @typeof-time: day  $\in$   $\mathbb{Z}$ 
42 @typeof-center_stock_amount: center_stock_amount  $\in$   $\mathbb{Z}$ 
43
44 @typeof-user-names: citizen_names  $\in$  NHS_NR  $\rightarrow$  NAME
45 @typeof-nhs-numbers: nhs_numbers  $\in$  USER  $\leftrightarrow$  NHS_NR
46 @typeof-certificates: certificates  $\in$  NHS_NR  $\leftrightarrow$  CERTIFICATE
47 @typeof-appointments-static: appointments_nr  $\in$  CENTER  $\leftrightarrow$   $\mathbb{Z}$ 
48
49 @typeof-citizen-name: citizen_name  $\subseteq$  NAME
50 @typeof-citizen-certs: citizen_certs  $\subseteq$  CERTIFICATE
51 @typeof-certificate-vaccines: certificate_vaccines  $\in$  CERTIFICATE  $\rightarrow$  BATCH_NR
52 @typeof-certificate-dates: certificate_dates  $\in$  CERTIFICATE  $\rightarrow$   $\mathbb{Z}$ 
53
54 events
55
56 /* Initialize new variables */
57 event INITIALISATION extends INITIALISATION
58 then
59   @act-init-vaccines: vaccines :=  $\emptyset$ 
60   @act-init-vaccines-count: vaccines_count :=  $\emptyset$ 
61   @act-init-centers: centers :=  $\emptyset$ 
62   @act-init-center_stock: center_stock :=  $\emptyset$ 
63   @act-init-center_info: center_vaccines :=  $\emptyset$ 
64   @act-init-day: day := 0
65   @act-init-center_stock_amount: center_stock_amount := 0
66   @act-init-names: citizen_names :=  $\emptyset$ 
67   @act-init-nhs: nhs_numbers :=  $\emptyset$ 
68   @act-init-certificates: certificates :=  $\emptyset$ 
69   @act-init-appointments: appointments_nr :=  $\emptyset$ 
70   @act-init-name: citizen_name :=  $\emptyset$ 
71   @act-init-certs: citizen_certs :=  $\emptyset$ 
72   @act-init-cert-vaccines: certificate_vaccines :=  $\emptyset$ 
73   @act-init-cert-dates: certificate_dates :=  $\emptyset$ 
74 end

```

```

75
76 /* Move time to the next day */
77 event NextDay
78 then
79   @inc-day: day := day + 1
80 end
81
82 /* Issue a vaccination certificate to the citizen. In the next machine it will be limited
83  * but now it is possible to issue whenever. */
84 event IssueCertificate
85 any
86   user
87   nhs
88   cert
89   batch
90 where
91   @check-staff: user ∈ dom(staff)
92   @check-loggedin: user ∈ dom(login)
93   @check-nhs: nhs ∈ ran(nhs_numbers)
94   @check-nhs-user: nhs_numbers ~ [{nhs}] ⊆ dom(users)
95   @check-cert: cert ∉ ran(certificates)
96   @check-vaccine: batch ∈ dom(vaccines_count)
97   @check-count: vaccines_count(batch) > 0
98 then
99   @act1: certificates := certificates ∪ {nhs ↦ cert}
100   @act2: certificate_vaccines := certificate_vaccines ∪ {cert ↦ batch}
101   @act3: certificate_dates := certificate_dates ∪ {cert ↦ day}
102   @act4: vaccines_count := vaccines_count ◁ {batch ↦ (vaccines_count(batch) - 1)}
103 end
104
105 /* Let staff and admins get the name of a user using their NHS number */
106 event GetCitizenName
107 any
108   user
109   nhs
110 where
111   @check-admin-or-staff: user ∈ dom(admins) ∨ user ∈ dom(staff)
112   @check-loggedin: user ∈ dom(login)
113   @check-nhs: nhs ∈ dom(citizen_names)
114 then
115   @act1: citizen_name := {citizen_names(nhs)}
116 end
117
118 /* Let the citizen get their certificates */
119 event GetCitizenCertificatesUser
120 any
121   user
122 where
123   @check-user: user ∈ dom(users)
124   @check-loggedin: user ∈ dom(login)
125 then
126   @act1: citizen_certs := certificates[{nhs_numbers(user)}]
127 end
128
129 /* Get citizen certificate with their MHS number */
130 event GetCitizenCertificatesNHS

```

```

131 any
132   user
133   nhs
134 where
135   @check-admin-or-staff: user ∈ dom(admins) ∨ user ∈ dom(staff)
136   @check-loggedin: user ∈ dom(login)
137   @check-nhs: nhs ∈ ran(nhs_numbers)
138 then
139   @act1: citizen_certs := certificates[{nhs}]
140 end
141
142 /* Add a vaccine to the system */
143 event AddVaccine
144 any
145   brand
146   batch
147   count
148 where
149   @grd1: brand ∈ BRAND
150   @grd2: batch ∉ dom(vaccines)
151   @grd3: count ∈ ℤ ∧ count ≥ 1
152 then
153   @act1: vaccines := vaccines ∪ {batch ↦ brand}
154   @act2: vaccines_count := vaccines_count ∪ {batch ↦ count}
155 end
156
157 /* Add a center to the system */
158 event AddCenter
159 any
160   user
161   center
162   appoint_count
163 where
164   @check-admin: user ∈ dom(admins)
165   @check-loggedin: user ∈ dom(login)
166   @check-center: center ∉ centers
167   @check-appoint-nr: appoint_count ∈ ℤ ∧ appoint_count ≥ 0
168 then
169   @add-center: centers := centers ∪ {center}
170   @add-center-appoint-count: appointments_nr := appointments_nr ∪ {center ↦ appoint_count}
171 end
172
173 /* Change the number of daily appointments for a center */
174 event ChangeCenterAppointCount
175 any
176   user
177   center
178   count
179 where
180   @check-admin: user ∈ dom(admins)
181   @check-loggedin: user ∈ dom(login)
182   @check-center: center ∈ centers
183   @check-count: count ∈ ℤ ∧ count ≥ 0
184 then
185   @act1: appointments_nr := appointments_nr ⋈ {center ↦ count}
186 end

```

```

187
188 /* Assign a batch to a vaccination center */
189 event AssignBatchToCenter
190 any
191   user
192   batch
193   center
194 where
195   @check-admin: user ∈ dom(admins)
196   @check-loggedin: user ∈ dom(login)
197   @check-batch-vaccine: batch ∈ dom(vaccines)
198   @check-batch-center: batch ∉ dom(center_stock)
199   @check-center: center ∈ centers
200 then
201   @act1: center_stock := center_stock ∪ {batch ↦ center}
202 end
203
204 /* Show vaccination center stock */
205 event ShowCenterStock
206 any
207   center
208 where
209   @check-center: center ∈ centers
210 then
211   @act1: center_vaccines := center_stock ∼ [{center}]
212 end
213
214 /* Initialize the root user as admin if there are no admins */
215 event InitRootUser extends InitRootUser
216 any
217   nhs
218   name
219 where
220   @check-nhs: nhs ∉ ran(nhs_numbers)
221   @check-name: name ∈ NAME
222 then
223   @add-citizen-name: citizen_names := citizen_names ∪ {nhs ↦ name}
224 end
225
226 /* Login a user */
227 event LoginUser extends LoginUser
228 end
229
230 /* Logout a user */
231 event LogoutUser extends LogoutUser
232 end
233
234 /* Register a new basic user */
235 event RegisterUser extends RegisterUser
236 any
237   nhs
238   name
239 where
240   @check-nhs: nhs ∉ dom(citizen_names)
241   @check-name-unique: name ∉ ran(citizen_names)
242 then

```



```

243  @add-nhs: nhs_numbers := nhs_numbers  $\cup$  {user  $\mapsto$  nhs}
244  @add-name: citizen_names := citizen_names  $\cup$  {nhs  $\mapsto$  name}
245  end
246
247  /* Change a user's password */
248  event ChangeUserPassword extends ChangeUserPassword
249  end
250
251  /* Change a staff's password */
252  event ChangeStaffPassword extends ChangeStaffPassword
253  end
254
255  /* Change an admin's password */
256  event ChangeAdminPassword extends ChangeAdminPassword
257  end
258
259  /* Elevate a user to staff role Their name and nhs number are removed.*/
260  event ElevateUserStaff extends ElevateUserStaff
261  then
262    @remove-name: citizen_names := nhs_numbers[{elevate_user}]  $\triangleleft$  citizen_names
263    @remove-nhs: nhs_numbers := nhs_numbers  $\setminus$  ({elevate_user}  $\triangleleft$  nhs_numbers)
264  end
265
266  /* Elevate a user to admin role. Their name and nhs number are removed. */
267  event ElevateUserAdmin extends ElevateUserAdmin
268  then
269    @remove-name: citizen_names := citizen_names  $\setminus$  (nhs_numbers[{user}]  $\triangleleft$  citizen_names)
270    @remove-nhs: nhs_numbers := nhs_numbers  $\setminus$  ({elevate_user}  $\triangleleft$  nhs_numbers)
271  end
272
273  /* Elevate staff member to admin role */
274  event ElevateStaffAdmin extends ElevateStaffAdmin
275  end
276
277  end

```

5 Task 2. Event-B model 1.3 Appointment booking and rebooking

This section pertains to appointment bookings. Here a citizen can book an appointment under certain conditions (as detailed in provided specifications). The requested criteria for how and under what conditions the citizen can make an appointment were respected and can be found under each requirement below:

REQ 19: *A citizen can only book an appointment if they have received less than 3 shots* - In the event StartBooking the cardinality of the citizen's certificates is checked to ensure that it is less than 3.

REQ 20: *A citizen can only book an appointment if their last shot was more than 28 days (4 weeks) from the date of login* - Using list comprehensions we go through the citizen's certificates and build a set of all the certificates that are older than 28 days from the day of certificate issue. If this set is the same as the citizen certificates set this means that all the citizen's certificates are older than 28 days and we proceed.

REQ 21: *A citizen is offered the earliest available appointment at any centre* - The event GetNewBookingSuggestion will put all the centres available for a day in the appointments_suggested set offers the earliest available appointment.

REQ 22: *If the citizen rejects the current offer, the system offers the following earliest appointment, that might be at a different centre* - The event RejectBookingSuggestion will increment the tmp_day that will move us to the next day and will make GetNewBookingSuggestion available again to query the centres available that day.

REQ 23: *If the citizen accepts the current offer, the booking is confirmed* - The event AcceptBookingSuggestion will assign a new unique appointment to the sets storing the association to a NHS number, center and the day of the appointment. This whole flow described above is controlled by a helper variable state that enables and disables the events when appropriate.

REQ 24: *At any point during the booking process, the use can abandon the system without completing the booking* - The event AbandonAppointmentBooking will set the day we want to book to 15. This will trigger the events described in REQ 25 and we will interrupt the appointment booking flow.

REQ 25: *The booking is stopped if there are no more available appointments within the 14 days from the date of login to offer* - All the events have guards that will stop the flow when the day we are checking for reaches 15 and the StartAppointmentRegistration event will become again available to start a new booking flow.

REQ 26: *A citizen can view their current booking (if any).* - GetCitizenBookings enables users to view their bookings that are stored in the citizen_books set which allows for them to view their current booking.

REQ 27: *Rebooking an appointment can be done up to 1 day before the date of the original*

one - The guard in StartRebooking makes sure of this by checking that there are appointments that have their day older than the current day using a list comprehension.

REQ 28 *Rebooking follows the same rules as booking* - The flow is practically the same for the StartRebooking, GetNewRebookingSuggestion, AcceptRebookingSuggestion and RejectRebookingSuggestion with small changes to update the sets related to appointments instead of adding to them and some additional guards.

REQ 29: *After re-booking is confirmed, the original appointment becomes available* - This occurs as there are separate sets storing the information of an appointment, updating it is enough to free the old appointment slot.

```
1 context AppointmentManagementContext extends VaccinationCentersContext
2 sets
3 APPOINTMENT
4 end
```

```
1 machine AppointmentManagement
2 refines VaccinationCenters
3 sees AppointmentManagementContext
4
5 /*AppointmentManagementContext
6 * Requirements implemented:
7 * REQ 18 A vaccination centre maintains a list of booked appointments
8 * REQ 19 A citizen can only book an appointment if they have received less than 3 shots
9 * REQ 20 A citizen can only book an appointment if their last shot was more than 28 days
10 * (4 weeks) from the date of the last shot
11 * REQ 21 A citizen is offered the earliest available appointment at any centre
12 * REQ 22 If the citizen rejects the current offer, the system offers the following earliest
13 * appointment, that might be at a different centre
14 * REQ 23 If the citizen accepts the current offer, the booking is confirmed
15 * REQ 24 At any point during the booking process, the use can abandon the system
16 * without completing the booking
17 * REQ 25 The booking is stopped if there are no more available appointments within the
18 * 14 days from the date of login to offer
19 * REQ 26 A citizen can view their current booking (if any).
20 * REQ 27 Rebooking an appointment can be done up to 1 day before the date of the original one
21 * REQ 28 Rebooking follows the same rules as booking
22 * REQ 29 After re-booking is confirmed, the original appointment becomes available
23 */
24
25 variables
26 users
27 admins
28 staff
29 login
30 register
31 vaccines
32 centers
33 center_stock
34 center_vaccines
```

```

35 day
36 day_tmp
37 citizen_names
38 nhs_numbers
39 certificates
40 appointments_nr
41 citizen_name
42 vaccines_count
43 citizen_certs
44 certificate_vaccines
45 certificate_dates
46 center_stock_amount
47
48 appointments_potential
49 appointments_suggested
50 calendar
51 state
52 user_session
53 appointment_nhs
54 appointment_center
55 appointment_day
56 citizen_books
57
58 invariants
59 @typeof-appointments-potential: appointments_potential  $\subseteq$  centers
60 @typeof-appointments-suggested: appointments_suggested  $\subseteq$  centers
61 @typeof-calendar: calendar  $\in$  centers  $\leftrightarrow \mathbb{Z}$ 
62 @typeof-day-tmp: day_tmp  $\in \mathbb{Z}$ 
63 @typeof-state: state  $\in \mathbb{Z}$ 
64
65 @typeof-user-session: user_session  $\subseteq$  USER
66 @typeof-appoint-nhs: appointment_nhs  $\in$  APPOINTMENT  $\leftrightarrow$  NHS_NR
67 @typeof-appoint-center: appointment_center  $\in$  APPOINTMENT  $\leftrightarrow$  CENTER
68 @typeof-appoint-day: appointment_day  $\in$  APPOINTMENT  $\leftrightarrow \mathbb{Z}$ 
69 @typeof-citizen-books: citizen_books  $\subseteq$  APPOINTMENT
70
71 events
72
73 /* Initialize new variables */
74 event INITIALISATION extends INITIALISATION
75 then
76   @act-init-appointments-offered: appointments_potential :=  $\emptyset$ 
77   @act-init-appointments-suggested: appointments_suggested :=  $\emptyset$ 
78   @act-init-calendar: calendar :=  $\emptyset$ 
79   @act-init-day-tmp: day_tmp := 0
80   @act-init-state: state := 0
81   @act-init-user-session: user_session :=  $\emptyset$ 
82   @act-init-appoint-nhs: appointment_nhs :=  $\emptyset$ 
83   @act-init-appoint-center: appointment_center :=  $\emptyset$ 
84   @act-init-appoint-day: appointment_day :=  $\emptyset$ 
85   @act-init-citizen-books: citizen_books :=  $\emptyset$ 
86 end
87
88 /* Issue certificate is now restricted to make certificates on the day of the appointment */
89 event IssueCertificate extends IssueCertificate
90 any

```

```

91  appointment
92  where
93    @check-appointment: appointment ∈ appointment_nhs ∼ [{nhs}]
94    @check-day: appointment_day(appointment) = day
95  end
96
97  /* Get citizen's appointments */
98  event GetCitizenBookings
99  any
100  user
101  where
102    @check-citizen: user ∈ dom(users)
103    @check-loggedin: user ∈ dom(login)
104  then
105    @act1: citizen_books := appointment_nhs ∼ [{nhs_numbers(user)}]
106  end
107
108  /* Starts the rebooking sequence */
109  event StartRebooking
110  any
111  user
112  where
113    /* Check the citizen is logged in */
114    @check-citizen: user ∈ dom(users)
115    @check-loggedin: user ∈ dom(login)
116
117    /* Check that the citizen has less than 3 vaccines */
118    @check-vaccine-count: card(certificates[{nhs_numbers(user)}]) < 3
119
120    /* Check the user had their last vaccine more than 28 days ago. The list comprehension returns
121     * all the certificates that are 28 days old then it compares it to the set of citizen certificates.
122     * If they are the same then all the vaccines are 28+ days old.
123     */
124    @check-vaccine-last: {cert | cert ∈ certificates[{nhs_numbers(user)}] ∧ (day - certificate_dates(cert) >
125                        28)}
126    = certificates[{nhs_numbers(user)}]
127
128    /* Check that there are centers with non zero stock and have available appointments */
129    @check-non-zero: {ctr | ctr ∈ {cr | cr ∈ centers ∧ {btch | btch ∈ center_stock ∼ [{cr}] ∧ vaccines_count(
130                        btch) > 0} ≠ ∅} ∧
131                      ((card({ctr} < calendar) < appointments_nr(ctr) ∨ ctr ∉ dom(calendar)) ∧ appointments_nr(ctr) ≠ 0)
132                      } ≠ ∅
133
134    @check-appointment: nhs_numbers(user) ∈ ran(appointment_nhs)
135    @check-day: {appoint | appoint ∈ appointment_nhs ∼ [{nhs_numbers(user)}] ∧ appointment_day(
136                  appoint) > day} ≠ ∅
137  then
138    @get-valid-centers: appointments_potential := {ctr | ctr ∈ centers ∧
139              {btch | btch ∈ center_stock ∼ [{ctr}] ∧ vaccines_count(btch) > 0} ≠ ∅
140            }
141    @init-first-day: day_tmp := day + 1
142    @change-state: state := 3
143    @save-user: user_session := {user}
144  end
145
146  /* Generate rebooking options for a day in the calendar. Make unavailable after 14 days from today. */

```

```

144 event GetNewRebookingSuggestion
145 where
146   @appointment-started: appointments_potential  $\neq \emptyset$ 
147   @check-day-tmp: day_tmp < day + 15
148   @check-state: state = 3
149 then
150   @get-valid-centers: appointments_suggested := {ctr | ctr  $\in$  appointments_potential  $\wedge$ 
151     ((card({ctr}  $\triangleleft$  calendar) < appointments_nr(ctr)  $\vee$  ctr  $\notin$  dom(calendar))  $\wedge$  appointments_nr(ctr)  $\neq 0$ )
152   }
153   @change-state: state := 4
154 end
155
156 /* Reject the rebooking options for that day. Move to the next day. Make unavailable after 14 days from
157    today.*/
157 event RejectRebookingSuggestion
158 where
159   @check-day-tmp: day_tmp < day + 15
160   @check-state: state = 4
161 then
162   @change-day: day_tmp := day_tmp + 1
163   @change-state: state := 3
164 end
165
166 /* Accept a rebooking option and change the booking information. Make unavailable after 14 days from
167    today.*/
167 event AcceptRebookingSuggestion
168 any
169   center
170   appointment
171   user
172 where
173   @check-center: center  $\in$  appointments_suggested
174   @check-appointment: appointment  $\notin$  dom(appointment_nhs)
175   @check-day-tmp: day_tmp < day + 15
176   @check-state: state = 4
177   @check-user-session: user  $\in$  user_session
178 then
179   @calendar-add: calendar := calendar  $\cup$  {center  $\mapsto$  day_tmp}
180   @link-center-change: appointment_center := appointment_center  $\triangleleft$  {appointment  $\mapsto$  center}
181   @link-day-change: appointment_day := appointment_day  $\triangleleft$  {appointment  $\mapsto$  day_tmp}
182   @interrupt-session: day_tmp := day + 15
183   @change-state: state := 0
184 end
185
186 /* Stop the rebooking sequence.*/
187 event AbandonAppointmentRebooking
188 where
189   @check-state: state = 3  $\vee$  state = 4
190 then
191   @interrupt-session: day_tmp := day + 15
192   @change-state: state := 0
193 end
194
195 /* Starts the booking sequence */
196 event StartAppointmentRegistration
197 any

```

```

198  user
199  where
200  /* Same as rebooking */
201  @check-citizen: user ∈ dom(users)
202  @check-loggedin: user ∈ dom(login)
203  @check-vaccine-count: card(certificates[{nhs_numbers(user)}]) < 3
204  @check-vaccine-last: {cert | cert ∈ certificates[{nhs_numbers(user)}] ∧ (day - certificate_dates(cert) >
205    27)} = certificates[{nhs_numbers(user)}]
206  @check-non-zero: {ctr | ctr ∈ {cr | cr ∈ centers ∧ {btch | btch ∈ center_stock ~ [{cr}]} ∧ vaccines_count(
207    btch) > 0} ≠ ∅} ∧
208    ((card({ctr} < calendar) < appointments_nr(ctr) ∨ ctr ∉ dom(calendar)) ∧ appointments_nr(ctr) ≠ 0)
209  } ≠ ∅
210  then
211  @get-valid-centers: appointments_potential := {ctr | ctr ∈ centers ∧ {btch | btch ∈ center_stock ~ [{ctr}]}
212    ∧ vaccines_count(btch) > 0} ≠ ∅}
213  @init-first-day: day_tmp := day + 1
214  @change-state: state := 1
215  @save-user: user_session := {user}
216  end
217  /* Generate booking options for a day in the calendar. Make unavailable after 14 days from today. */
218  event GetNewAppointmentSuggestion
219  where
220  @appointment-started: appointments_potential ≠ ∅
221  @check-day-tmp: day_tmp < day + 15
222  @check-state: state = 1
223  then
224  @get-valid-centers: appointments_suggested := {ctr | ctr ∈ appointments_potential ∧
225    ((card({ctr} < calendar) < appointments_nr(ctr) ∨ ctr ∉ dom(calendar)) ∧ appointments_nr(ctr) ≠ 0)
226    }
227  @change-state: state := 2
228  end
229  /* Reject the booking options for that day. Move to the next day. Make unavailable after 14 days from
230  today. */
231  event RejectAppointmentSuggestion
232  where
233  @check-day-tmp: day_tmp < day + 15
234  @check-state: state = 2
235  then
236  @init-first-day: day_tmp := day_tmp + 1
237  @change-state: state := 1
238  end
239  /* Accept a booking option and change the booking information. Make unavailable after 14 days from
240  today. */
241  event AcceptAppointmentSuggestion
242  any
243  center
244  appointment
245  user
246  where
247  @check-center: center ∈ appointments_suggested
248  @check-appointment: appointment ∉ dom(appointment_nhs)
249  @check-day-tmp: day_tmp < day + 15
250  @check-state: state = 2

```

```

249  @check-user-session: user ∈ user_session
250  then
251  @calendar-add: calendar := calendar ∪ {center ↦ day_tmp}
252  @link-citizen: appointment_nhs := appointment_nhs ∪ {appointment ↦ nhs_numbers(user)}
253  @link-center: appointment_center := appointment_center ∪ {appointment ↦ center}
254  @link-day: appointment_day := appointment_day ∪ {appointment ↦ day_tmp}
255  @interrupt-session: day_tmp := day + 15
256  @change-state: state := 0
257  end
258
259  /* Stop the booking sequence.*/
260  event AbandonAppointmentRegistration
261  where
262  @check-state: state = 1 ∨ state = 2
263  then
264  @interrupt-session: day_tmp := day + 15
265  @change-state: state := 0
266  end
267
268  event NextDay extends NextDay
269  end
270
271  event GetCitizenName extends GetCitizenName
272  end
273
274  event GetCitizenCertificatesUser extends GetCitizenCertificatesUser
275  end
276
277  event GetCitizenCertificatesNHS extends GetCitizenCertificatesNHS
278  end
279
280  event AddVaccine extends AddVaccine
281  end
282
283  event AddCenter extends AddCenter
284  end
285
286  event ChangeCenterAppointCount extends ChangeCenterAppointCount
287  end
288
289  event AssignBatchToCenter extends AssignBatchToCenter
290  end
291
292  event ShowCenterStock extends ShowCenterStock
293  end
294
295  event InitRootUser extends InitRootUser
296  end
297
298  event LoginUser extends LoginUser
299  end
300
301  event LogoutUser extends LogoutUser
302  end
303
304  event RegisterUser extends RegisterUser

```



```
305 end
306
307 event ChangeUserPassword extends ChangeUserPassword
308 end
309
310 event ChangeStaffPassword extends ChangeStaffPassword
311 end
312
313 event ChangeAdminPassword extends ChangeAdminPassword
314 end
315
316 event ElevateUserStaff extends ElevateUserStaff
317 end
318
319 event ElevateUserAdmin extends ElevateUserAdmin
320 end
321
322 event ElevateStaffAdmin extends ElevateStaffAdmin
323 end
324
325 end
```